
Referat: 
 
Ziel der Bachelorarbeit ist es, ein Festwalzprozess im FE-Programm ANSYS zu 
simulieren. Dazu muss das Modell zuerst gebaut werden. Danach wird der 
Simulationstyp definiert. Am Ende können die Ergebnisse der Simulation 
bestimmt und analysiert werden. 

1. Aufgabenstellung 
 

1.1 Grundlagen des Festwalzens 

Das Festwalzen ist das einzige Randschicht- verfestigungsverfahren, welches eine 
Glättung der Oberfläche, die Induzierung von Eigenspannungen und eine 
Kaltverfestigung bewirkt. Hierdurch ergibt sich eine außerordentliche Steigerung 
der Schwingfestigkeit. Der Anwendungsbereich des Festwalzens erstreckt sich 
nahezu über das gesamte Spektrum der metallischen Werkstoffe.  
 
Im Zusammenspiel mit der Verringerung der Rautiefe und der Kaltverfestigung 
ergeben sich durch die Anwendung des Festwalzens spürbare Steigerungen der 
Lebensdauer eines Bauteils. [1] 
 

1.2 Übersicht der Aufgabe 

Anhand der in Abb.1 dargestellten Geometrie einer Rundprobe soll das 
Festwalzen der Radien R im Einstichverfahren für die ADI-Gusswerkstoffe EN 
GJS-800-8 und EN GJS-1000-5 mit ANSYS simuliert werden. 
 

 
Abb.1:Geometrie der Probe und der Walzrolle 

1.3 Parameter von der Aufgabe 

 

● Konstruktive Parameter 
  -  Probendurchmesser d:       d1 = 25mm 
                             d2 = 30mm 
 
  -  Radius R:                 R1 = 1,1mm 

R2 = 1,5mm 
    

● Fesrwalzparameter 
  -  Rollendurchmesser DR:          DR = 15mm 
 
  -  Profilradius der Walzrolle rR:       für R1 → rR1 = 1,1mm 
                                   für R2 → rR2 = 1,45mm 
 

  -  Anstellwinkel der Walzrolle αw:   αw1 = 0° 

 
  -  Festwalzkraft Fw:                Fw1 = 6kN        Fw3 = 10kN 
                                   Fw2 = 8kN        Fw4 = 12kN 
    

● Werkstffkenngrößen 
- Fließkurven zur Charakterisierung des Wekstoffverhaltens von EN 

GJS-800-8 und EN GJS-1000-5 
 

Tabelle 1: Wekstoff-Einflussgrößen 
 

Werkstoff E-Modul E [GPa] Poisson’sche Konstante v 

EN GJS-800-8 170 2,7 
EN GJS-1000-5 168 2,7 
Stahl (Walzrolle) 210 3,0 

1.4 Teilaufgaben 

 
1. Erstellung eines dreidimensionalen FE-Modells von der Probe mit der 
Walzrolle 
 
2. Simulation der verschiedenen Beanspruchungsfälle durch Variation der 
konstruktiven Parameter (d, R) und der Festwalzparameter (Fw, αw) für beide 
Werkstoffe 
 
3. Ermittlung der axialen, radialen und tangentialen Eigenspannungen (Betrag, 
Verteilung) für den jeweiligen Beanspruchungsfall in Form eines Tiefenprofils 
 
4. Ermittlung der Einwalztiefe (plastische Dehnung) für den jeweiligen 
Beanspruchungsfall 
 
5. Welchen Einfluss haben der Probendurchmesser d auf den Betrag bzw. die 
Verteilung der Eigenspannungen? 
 
6. Bei welcher Festwalzkraft Fw treten für den jeweiligen ADI-Werkstoff und 
Radius R die höchsten Druckeigenspannungen auf? 

2. Modellierung 

2.1 Werkstück 

2.1.1 Methode der Modellierung 

● Der Schnitt durch die Probe ist symmetrisch zur Mittellinie. Um einfach das 
Modell zu analysieren, wird also zuerst nur die Hälfte auf der XY-Ebene in 
ANSYS gezeichnet. Danach wird das Volumen mit dem Befehl‚VROTAT’ 
erstellt. 
 
● Es gibt einige Abschnitte der Probe, die bei der Beanspruchung im 
Festwalzprozess nicht beeinflusst werden. D.h. die Darstellung dieser Abschnitte 
im Modell ist nicht notwendig. Dadurch wird die Rechenzeit erheblich verkürzt. 

 
Abb.2: Komplette Schnittfläche in XY-Ebene 

 

                                        ↓ 

 
Abb.3: Vereinfachte Schnittfläche in XY-Ebene 
 

 

2.1.2 Die Positionen der Keypoints 

 
Abb.4: Keypoints in der Probe 

Tabelle 2: Die Positionen der Keypoints von der Probe im CS 0 
 

Nummer von KP X Y 
KP  1 2.5*d d/2 
KP  2 3.5*d-R-b d/2 
KP  3 3.5*d-R d/2-0.4 
KP  4 3.5*d d/2-0.4+R 
KP  5 3.5*d 1.4*d/2 
KP  6 4.5*d 1.4*d/2 
KP  7 4.5*d 0 
KP  8 2.5*d 0 

*KP  20 3.5*d-R d/2-0.4+R 
 (*Keypoint 20 ist nicht in der Abbildung. Es ist der Mittelpunkt der Kurve von 

Keypoint 2 bis Keypoint 4) 

 

2.2 Werkzeug 

2.2.1 Methode der Modellierung 

Der Schnitt durch die Walzrolle ist auch symmetrisch zur Mittellinie 
(Rotationsachse), deshalb wird auch nur die Hälfte auf der XY-Ebene in ANSYS 
gezeichnet. Dann wird das Volumen mit dem Befehl ‚VROTAT’ erstellt. 
 

2.2.2 Die Position des lokalen Koordinatensystems 

● Um einfach die Positionen der Keypoints zu bestimmen, kann ein lokales 
Koordinatensystem zuerst im Mittelpunkt der Walzrolle erstellt werden.   

 
Abb.5 :die Position von KP_3 und KP_30 

In Abb.5 können wir sehen, dass der Ursprung des lokalen Koordinatensystems 
11(CS_11) in der Position von Keypunkt_30(KP_30) sein soll. Im realen 
Festwalzprozess soll der Mittlpunkt(KP_3) der Kurve von Kerbe und der 
Mittelpunkt von der Rolle in einer geraden Linie sein. Deshalb ist es einfach, die 
Position von KP_30 zu bestimmen, wenn die Position von KP _3 schon bekannt 
ist. 
 
Die Position von KP _3:        X(KP_3) = 3.5*d-R  (Gl.1) 
                            Y(KP_3) = d/2-0.4   (Gl.2) 
 
 
Dann ist die Position von KP _30:  X(KP_30) = X(KP_3) = 3.5*d-R  (Gl.3) 
                              Y(KP_30) = Y(KP_3) + DR/2 = 
d/2-0.4+DR/2 (Gl.4) 
 
Die Position von CS _11 ist gleich wie KP _30 
 
X(CS_11) = 3.5*d-R       (Gl.5) 
Y(CS_11) = d/2-0.4+DR/2       (Gl.6) 
 

2.2.3 Die Position der Keypoints der Rolle im CS11 

 
Mit dem Befehl ,CSYS 11’ können die Keypoints in der Schnittfläche der Rolle 
im aktiven Koordinatensystem CS_11 schneller erstellt werden. 

                 
Abb.6: CS11                    Abb.7: KP in der Schnittfläche der Rolle 

Tabelle 3: Die Positionen der Keypoints von der Rolle im CS 11 
 

Nummer von KP X Y 
KP 30 0 0 
KP 31 0 rR 
KP 32 DR/2-rR rR 
KP 33 DR/2 0 
KP 34 DR/2-rR -rR 
KP 35 0 -rR 
*KP 36 DR/2-rR 0 

 

(*Keypoint 36 ist nicht in der Abbildung. Es ist der Mittelpunkt der Kurve von Keypoint 

32 bis Keypoint 34) 

3. Vernetzung 

 

3.1 Material und Elementtyp 

 

3.1.1 Material 

Die Technischen Spannungs- und Dehnungswerte der zwei Materialen sind schon 
gegeben. Durch die Gleichungen (Gl.7) und (Gl.8) kann die Technische 
Spannungs- und Dehnungswerte in wahre Spannungs-Dehnungswerte übertragen 
werden. 
 
Umrechnung technische/wahre Werte: 

 

(1 )w t tσ σ ε= +   (Gl.7) 

ln(1 )w tε ε= +   (Gl.8) 

 
 
 
·Die Wahre Spannungs-Dehnungswerte von EN GJS-800-8 

 
Abb.8 

Die Streckgrenze ist ca. 650 MPa 

● Die Wahre Spannungs-Dehnungswerte von EN GJS-1000-5 

 

Abb.9 
Die Streckgrenze ist ca. 900 MPa 

 
Die Daten der zwei Materialen werden in die Unterprogramme ADI_800.inp und 
ADI_1000 geschrieben, damit diese im Hauptprogramm von ANSYS aufgerufen 
werden können.  
 

3.1.2 Elementtyp 

1.Elementtyp : SOLID 45 
 
Elementtyp für Probe: 1.Elementtyp 
Elementtyp für Rolle : 1.Elementtyp 
 
 
Obwohl die Zahl von DOF von SOLID 45 gering ist, hat das Ergebnis noch eine 
hohe Genauigkeit, wenn die Modelle kartografisch vernetzt werden. Außerdem 
kann die Rechenzeit reduziert werden. 

3.2 Vernetzung des Werkstücks 

Um die Rechenzeit zu reduzieren, soll die Zahl von DOF reduziert werden. Es ist 
besser, das Modell kartografisch zu vernetzen. Das Volumen ist nicht regelrecht 
sichtbar. Deshalb soll das Volumen in einige regelrechte Volumina geteilt werden. Die 
Schnittfläche kann zuerst in einige regelrechte Flächen geteilt werden, dann werden 
die Volumen auch regelrecht sein, welche in der Schnittfläche mit ,VROTAT’ erstellt 
werden. 

 
Abb.10: verteilte Schnittfläche der Probe 

 
In Abb.12 :   x1=0.1*d      y1=d/2-x1 
 

Tabelle 4: Die Positionen der Keypoints von der Verteilung im CS 0 
 

Nummer von KP X Y 
KP  9 3.5*d-R-b 0 
KP  10 3.5*d+x1 0 
KP  11 3.5*d+x1 d/2-0.4+R 
KP  12 3.5*d-R-b y1 
KP  13 5*d d/2-0.4+R 
KP  14 3.5*d+x1 1.4*d/2 
KP  15 2.5*d y1 

 
Weil der Festwalzprozess eine Kontakt-Aufgabe darstellt, sollen die Kontakt-Fläche 
und die Target-Fläche fein vernetzt werden. Deshalbe muss die Fläche in der Kerbe 
der Probe fein vernetzt werden. Mit dem Befehl ,LESIZE’ kann die Kurve der Kerbe 
fein vernetzt werden. 
 
Nach der Erstellung der Volumina kann die ganze Probe einfach beim 1.Materialtyp 
und 1.Elementtyp vernetzt werden. 

Abb.11: vernetzte Schnittfläche der Probe 
 
 

 
Abb.12: vernetzte Probe 

3.3 Vernetzung des Werkzeugs 

Die Methode der Vernetzung der Walzrolle ist gleich wie die für die Probe. 

                     
Abb.13:verteilte Schnittfläche der Rolle           Abb.14:vernetzte Schnittfläche der Rolle 

 
Die Position von KP 37 von der Verteilung im CS 11: 
X(KP 37)=DR/2-2*rR  ,  Y(KP 37)=0 
 
Die Fläche im Profil der Rolle muss fein vernetzt werden. Mit dem Befehl ,LESIZE’ 
kann die Kurve des Profils fein vernetzt werden. 
 
Nach der Erstellung des Volumens, kann die ganze Rolle einfach beim 2.Materialtyp 
und 1.Elementtyp vernetzt werden. 

 
Abb.15: vernetzes Modell 

4. Definition vom Kontakt 

 

4.1 Kontakttyp 

 
● elastisch zu elastisch  

—Das Werkstück ist elastisch-plastisch 

—Die Walzrolle ist rein elasctisch 
    

● Fläche zur Fläche 
 

— “Contact”-Fläche:    Die Oberfläche der Kerbe vom Werkstück 
— “Target”-Fläche:     Die Oberfläche des Profils vom Werkstück 
 

— Elementtyp (ET):       2.ET “CONTAC173” für “Contact”-Fläche 
                        3.ET “TARGET170” für “Target”-Fläche 

 

4.2 Real-Definition 

 
“CONTAC173” und “TARGET170” gehören zu einer Kontakt-Gruppe, deshalb soll 
das “Real” dieser zwei Elementtypen gleich sein. Das “Real” ist in dieser Aufgabe 
“Real 3”. 
 
Definition des “Real 3”: 
 
R,3,,,0.5,0.2,0.001,,,,(SIGMAF*0.577),0.0,1.0,,    
    ! 1 - R1 (intern) 
    ! 2 - R2 (intern) 
    ! 3 - FKN - Kontaktsteifigkeit (default: 1.0) 
    ! 4 - FTOL (default: 0.1)  
    ! 5 - ICONT - initial closure  
    ! 6 - PINB 
    ! 7 - PMIN 
    ! 8 - PMAX 
    ! 9 - TAUMAX (SIGMAF*1/sqrt(3)) 
    !10 - CNOF - contact offset 
    !11 - FKOP 
    !12 - FKT 
    !13 - COHE sliding resistance (0) 
 
* SIGMAF ist der Parameter der Streckgrenze des Materials vom Werkstück. 

4.3 “Element Key”-Definition  

 
Außer der Real-Definition soll “Element Key” noch definiert werden. Damit ist die 
Definition des Kontaktproblems komplett und exakt. 
 
Definition des “Element Key” von 3.ET “TARGET170” : 
 
 
KEYOPT,3,1,     ! 1 - only structural contact 
KEYOPT,3,2,0  ! 2 - Algorithmus: 0 - Penalty function + Lagrange multiplier  
KEYOPT,3,3,0   ! 3 - Use with h-elements (no superelements) 
KEYOPT,3,4,0    ! 4 - Location of contact detection point: 0 - Gauss-Point  
KEYOPT,3,5,     ! 5 - CNOF Automated adjustment 
KEYOPT,3,6,     ! 6 - Symmetrical stiffness matrix 
KEYOPT,3,7,0   ! 7 - Element level time inc. control.; 0 - No control 
KEYOPT,3,8,0    ! 8 - Suprious Contact Prevention 0 - no prevention 
KEYOPT,3,9,0   ! 9 - include both init. - geom. penet. or gap and offset 
KEYOPT,3,10,0   !10 - Contact stiffness update: 0 - no update 
KEYOPT,3,11,0   !11 - Beam/Shell Thickness Effect: 0 - None  
KEYOPT,3,12,0   !12 - Behavior of contact surface: 0 – Standart 

4.4 Erstellung der Kontakt-Gruppe 

 

● Da der Kontakttyp elastisch zu elastisch ist, sollen “CONTAC173” und 
“TARGET170” mit dem Befehl “ESURF” auf die Oberfläche der Kerbe und des 
Profils gelegt werden. 
    

● Die Walzrolle rollt nur 30° ab. Deshalb werden die Teiloberflächen von der Rolle 
und der Kerbe der Probe gebraucht, um die Kontaktelement zu positionieren. 
 

               

Abb.16: Oberfläche der Kerbe und der Walzrolle     Abb.17: Erstellte Kontakt- und Targetfläche 

5. Randbedingungen 

 
Abb.18: realer Festwalzprozess [2] 
    

● In Abb.20 können wir sehen, dass das Probenende eingespannt werden sollen. 
In der Aufgabe ist das Modell vereinfacht und nicht komplett dargestellt. Deshalb 
wird die Schnittfläche am Probenende nur in X Richtung eingeschränkt. 

Abb.19: Randbedingung am Probenende 

 

● Um symmetrische Eigenschaften zu erhalten, sollen die zwei Schnittflächen des 
Viertelmodells in Y – und Z- Richtung eingeschränkt werden. 

Abb.20: Symmetrierandbedingung 

6. Rechnungen der Bewegungen 

 

6.1 Beschreibung der Bewegungen des Festwalzprozesses 

 
Der Festwalzprozess besteht aus drei verschieden Bewegungen: 
 
Erste Bewegung: Die Walzrolle dringt in die Probe. 
Zweite Bewegung: Die Walzrolle rollt auf der Probe ab. 
Letzte Bewegung: Die Walzrolle ist weg von der Probe. 
 

6.2 Methode der Lösung in ANSYS 

    

● In ANSYS können wir finden, dass die Bahnen der drei Bewegungen nur in der 
YZ-Ebene sind. Deshlab kann diese Bewegungsaufgabe zuerst in eine 2D Aufgabe 
vereinfacht werden. Dann können die Bewegungsgleichungen in eine 3D Aufgabe 
übertragen werden. 
    

●Die Achse in der Mittel der realen Walzrolle (Abb.21) kann die Bewegungen der 
Rolle steuern. In ANSYS können fünf Knoten in der Mitte der Rolle ausgewählt 
werden, um die Achse zu ersetzen. 

 

Abb.21: reale Walzrolle [2] 

 

● Durch den Befehl “D” können die Knoten verschoben werden. 
    

● Die Abrollbewegung ist mit dem Schleifenbefehl “*DO...*ENDO” zu realisieren. 
Diese Bewegung soll in k Schritte geteilt werden. In jedem Schritt rotiert die Rolle um 
die Probe um Grad (Φ_Ink). Der gesamte Rotationsgrad ist k×Φ_Ink. 

6.3 Bewegungsrechungen für 2D-Schnittmodell 

      
Abb.22: vereinfachte 2D-Modell                    Abb.23: gewählte 5 Steuerknoten 
Probenradius R1                              Distanz zwischen 0 und 1=rr 
Rollenradius R2                              Distanz zwischen 0 und 2=rr 
R3=R1+R2     (Gl.9)                        Distanz zwischen 0 und 3=rr 
                                           Distanz zwischen 0 und 4=rr 
                                

6.3.1 Erste Bewegung: Eindringen 

 Abb.24: Die Walzrolle dringt in dieProbe 
R4=R1+R2-t   (Gl.10) 

● Verschiebung von Knoten 0 

 Abb.25: Die Bewegung von Knoten 0 im 1.Schritt 

 (X - und Y - Achse gehören zum Knotenkoordinatensystem in Abb.25.) 

 

Parameter: 
Ur0=R4 
 
Verschiebungswerte: 
In X-Achse:  UX = Ur0-R3 = - t    (Gl.11) 
In Y-Achse:  UY = 0 
 
Die Verschiebung von Knoten 1 und 2 sind gleich wie die Verschiebung des Knoten 0, 
damit können die Abbildungen ausgelassen werden. 
 
● Verschiebung von Knoten 1 
 
Parameter: 
Ur1=R4-rr 
 
Verschiebungswerte: 
In X-Achse:  UX = Ur1-(R3-rr) = - t    (Gl.12) 
In Y-Achse:  UY = 0 
 
● Verschiebung von Knoten 2 
Parameter: 
Ur2=R4+rr 
 
Verschiebungswerte: 
In X-Achse:  UX = Ur1-(R3+rr) = - t    (Gl.3) 
In Y-Achse:  UY = 0 

● Verschiebung von Knoten 3 

 
Abb.26 

Parameter: 
 
Ψ= arctan(rr/R4)-arctan(rr/R3)  (Gl.14) 

2 23 4rrUr R= +   (Gl.15) 

 
Verschiebungswerte: 
 

In X-Achse: 2 2 = Ur3 rrU s 3X Co RΨ − +i   (Gl.16) 

In Y-Achse:  UY = Ur3·SinΨ  (Gl.7) 

● Verschiebung von Knoten 4 

 
Abb.27 

Parameter: 
 
Ψ ist gleich wie (Gl.14) 

4Ur  ist gleich wie (Gl.15) 
 
Verschiebungswerte: 
 
In X-Achse: UX  ist gleich wie (Gl.16) 
In Y-Achse:  UY  ist gleich wie (Gl.17) 

6.3.2 Zweite Bewegung: Abrollen 

 
● Verschiebung von Knoten 0 

 
Abb.28 

Parameter: 
 
Φ0 = Φ_Ink·k  (Gl.18) 
α= (R4-R2)·Φ0/R2  (Gl.19) 
Ur0=R4 
 
Verschiebungswerte: 
 

In X-Achse: = Ur0 Cos(UX 0)-R3Φi   (Gl.20) 

In Y-Achse:  =  - Ur0 Sin )UY ( 0Φi   (Gl.21) 

● Verschiebung von Knoten 1 

 
Abb.29 

Parameter: 
 
θ1 = Φ0 –α  (Gl.22) 

2 2rr +Ur0 -2 rr Ur0 C1 os( 1)Ur θ= i i i   (Gl.23) 

2 2 2Ur1 +Ur0 -rr
=  arccos( )

2 Ur r
1 

1 U 0
β

i i

  (Gl.24) 

Φ1 = Φ0 –β1  (Gl.25) 
 
Verschiebungswerte: 
 

In X-Achse: = Ur1 os( 1)-(R3UX -rr)C Φi   (Gl.26) 

In Y-Achse: =  - Ur1 Sin )UY ( 1Φi   (Gl.27) 

● Verschiebung von Knoten 2 

 
Abb.30 

Parameter: 
 
θ2 = 180°-θ1  (Gl.28) 

2 2rr +Ur0 -2 rr Ur0 C2 os( 2)Ur θ= i i i   (Gl.29) 

2 2 2Ur2 +Ur0 -rr
=  arccos( )

2 Ur r
2 

2 U 0
β

i i

  (Gl.30) 

Φ2 = Φ0 +β2  (Gl.31) 
 
Verschiebungswerte: 
 

In X-Achse: = Ur2 os( 2)-(R3UX +rr)C Φi   (Gl.32) 

In Y-Achse: =  - Ur2 Sin )UY ( 2Φi   (Gl.33) 

● Verschiebung von Knoten 3 

 Abb.31 
Parameter: 
 
θ3 = 90°-θ1  (Gl.34) 

2 2r3_ r R30 +Ur =   (Gl.35) 

2 2rr +Ur0 -2 rr Ur0 C3 os( 3)Ur θ= i i i   (Gl.36) 

2 2 2Ur3 +Ur0 -rr
=  arccos( )

2 Ur r
3 

3 U 0
β

i i

  (Gl.37) 

arctan( )
3

rr

R
=ϒ   (Gl.38) 

Φ3 = Φ0 –β3 +γ  (Gl.39) 
 
Verschiebungswerte: 
 

In X-Achse: = Ur3 os( 3)-UUX r3_0C Φi   (Gl.40) 

In Y-Achse: =  - Ur3 Sin )UY ( 3Φi   (Gl.41) 



 26 

● Verschiebung von Knoten 4 

 Abb.32 
Parameter: 
 
θ4 = 90°+θ1  (Gl.42) 

2 2r4 _ r R30 +Ur =   (Gl.43) 

2 2rr +Ur0 -2 rr Ur0 C4 os( 4)Ur θ= i i i   (Gl.44) 

2 2 2Ur4 +Ur0 -rr
=  arccos( )

2 Ur r
4 

4 U 0
β

i i

  (Gl.45) 

arctan( )
3

rr

R
=ϒ   (Gl.46) 

Φ4 = Φ0 +β4 –γ  (Gl.47) 
 
Verschiebungswerte: 
 

In X-Achse: = Ur4 os( 4)-UUX r4_0C Φi   (Gl.48) 

In Y-Achse: =  - Ur4 Sin )UY ( 4Φi   (Gl.49) 

6.3.3 Letzte Bewegung: Verlassen 

→  

Abb.33: Letzter Schritt der Abrollbewegung            Abb.34: Rolle weg von der Probe 
    

● In dieser Bewegung gibt es nur einen Schritt. Die Werte der Parameter (Φ0 , α, θ1, 
θ2, θ3, θ4) bleiben wie die Werte im letzten Schritt der Abrollbewegung. 
    

● Da sich der Wert von Ur0 verändert, werden die restlichen Parameter neu bestimmt. 
Die Gleichungen für die Berechnung der neuen Werte sind gleich wie die in der 
Abrollbewegung. 

Für Knoten 0:   In X-Achse:   = Ur0' Cos(U R3X 0)-Φi         (Gl.50) 

In Y-Achse:    =  - Ur0' Si )UY n( 0Φi           (Gl.51) 

Für Knoten 1:   In X-Achse:   = Ur1' os( 1)-(R3UX -rr)C Φi       (Gl.52) 

In Y-Achse:    =  - Ur1' Si )UY n( 1Φi           (Gl.53) 

Für Knoten 2:   In X-Achse:   = Ur2' os( 2)-(R3UX +rr)C Φi      (Gl.54) 

In Y-Achse:   =  - Ur2' Si )UY n( 2Φi            (Gl.55) 

Für Knoten 3:   In X-Achse:   = Ur3' os( 3)-UUX r3_0C Φi       (Gl.56) 

In Y-Achse:   =  - Ur3' Si )UY n( 3Φi            (Gl.57) 

Für Knoten 4:   In X-Achse:   = Ur4' os( 4)-UUX r4_0C Φi       (Gl.58) 

In Y-Achse:   =  - Ur4' Si )UY n( 4Φi            (Gl.59) 

6.4 Bewegungsrechungen für 3D-Modell 

 
Zuerst wird ein lokales zylindrisches Koordinatensystem (CS12) in der Mitte der 
Probe erstellt. Dann werden fünf neue Knoten ausgewählt, welche in der XY-Ebene 
von CS12 bleiben. 
 
Die Gleichungen für die zeitabhängigen Randbedingungen aus 2D-Schnittmodell 
müssen nur auf die "neuen" Knoten übertragen werden. 

  

Abb.35: Position von CS12                    Abb.36: Schnittfläche in XY-Ebene von CS12 

Nummer der neuen Knoten: 
 
Knoten 2439 ersetzt Knoten 0 
Knoten 2444 ersetzt Knoten 1 
Knoten 3049 ersetzt Knoten 2 
Knoten 3583 ersetzt Knoten 3 
Knoten 2449 ersetzt Knoten 4 
 

Definition der Parameter im 3D-Modell: 
 
R1=d/2-0.4 
R2=DR/2 
rr = Distanz zwischen 1925 und 1930(1935,2535 oder 3069) 
Φ_Ink = 5° 
K=30°/Φ_Ink =6 (6 Schritte in Abrollbewegung) 

7. Berechnung der Eindringkraft 
 
Wegen der Begrenzung einiger Funktionen in ANSYS, wird die Eindringkraft nicht 
direkt an die Rolle übertragen. Aber durch die gegebene Verschiebung an die Rolle 
kann der gleiche Effekt erzielt werden. Deshalb muss die Beziehung zwischen 
Eindringkraft und Eindringtiefe bekannt sein. 
 

● Summation der Kontaktkräfte:  

Da die Verschiebungsrandbedingungen an den fünf Knoten gegeben sind, müssen nur 
die Reaktionskräfte gemessen werden. Danach kann die gesamte Reaktionskraft 
berechnet werden. Die gesamte Reaktionskraft in Y-Richtung vom originalen 
Koordinatensystem (CS 0) ist gleich der Eindringkraft. 
 

   

Abb.37: Reaktionskräfte                          Abb.38 
 
Eindringskraft: 
FR= (Fj0+Fj1+Fj2+Fj3_x*cosγ+Fj4_x*cosγ+Fj3_y*sinγ+ Fj4_y*sinγ)*2/1000  (Gl.60) 

● Vermessung der Y-Verschiebungen: 

Ein Knoten aus der Rolle wird gewählt. Mit dem Befehl “*get” kann die 
Y-Verschiebung vom Knoten gemessen werden. 
 

 Abb.39 

● Daten ausgeben 

Mit einem Post-Processing-Programm kann die Daten in einem Vektor gespeichert 
und als Textdatei ausgegeben werden. Diese Textdaten können in Excel verarbeitet 
werden. 
 
Mit verschiedenen Parametern ergeben sich unterschiedliche Verläufe der 
Eindringkraft über der Eindringtiefe. 
z.B für  d1, R1, EN GJS-800-8 

 
Abb.40  

für d1, R1, EN GJS-1000-5 

 
                              Abb.41 

● Gebrauchte Daten auswählen 

Tabelle 5 : EN GJS-800-8 
 
 Fw1 Fw2 Fw3 Fw4 

D1,R1 t=0.175 t=0.212 t=0.255 t=0.300 
D1,R2 t=0.165 t=0.195 t=0.230 t=0.260 
D2,R1 t=0.170 t=0.200 t=0.225 t=0.270 
D2,R2 t=0.165 t=0.180 t=0.210 t=0.240 

 
 

Tabelle 6 : EN GJS-1000-5 
 
 Fw1 Fw2 Fw3 Fw4 

D1,R1 t=0.170 t=0.200 t=0.225 t=0.265 
D1,R2 t=0.162 t=0.180 t=0.210 t=0.230 
D2,R1 t=0.165 t=0.191 t=0.210 t=0.230 
D2,R2 t=0.162 t=0.182 t=0.205 t=0.210 

8. Vermessung der Eigenspannungen 

8.1 Aufgabe 1 

Um den Effekt des Festwalzprozesses aufzuzeigen, sollen die Spannungen für alle 
Beanspruchungsfälle beim Abrollvorgang bis 30° an der Oberfläche der Kerbe 
gemessen werden. Die Spannungen, die nach dem Verlassen der Walzrolle noch in 
der Probe verbleiben , werden als Eigenspannungen bezeichnet. 
 

● Koordinaten für anzuzeigendes Ergebnis 

Für die Darstellung günstig der Spannungen und Verformungen ist die Anwendung 
eines zylindrischen Koordinatensystems günstig. Deshalb kann CS_12 als 
Auswertungskoordinatensystem mit dem Befehl “RSYS,11” aktiviert werden. 
 

● Zu vermessenden Knoten auswählen 

In der XY-Ebene von CS_12 kann ein zu vermessender Knoten ausgewählt werden. 
Manchmal ist der gewählte Knoten nicht in XY-Ebene, aber der Fehler des Ergebnises 
ist so klein, dass er vernachlässigt werden kann. 
 
Die Position vom gewählten Knoten (GN) im CS_12 ist (d/2-0.4,75,0). 

 

Abb.42: gewählter Knoten und die Spannungsrichtungen 

● Vermessung der Spannungen 

Mit dem Befehl “*Get” können die Spannungen des Knotens in drei Richtungen 
bestimmt werden. SX ist die radiale Spannung, SY die tangentiale Spannung und SZ 
die axiale Spannung.  

● Daten ausgeben 

Die Daten können in Excel verarbeitet werden. Für verschiedene Parameter ergeben 
sich verschidene Spannungsverläufe. Es gibt insgesamt 32 Beanspruchungfälle. 
 
Zum Beispiel: 
Gruppe 1: d1, R1, Fw1, EN GJS-800-8 

 
Abb.43 

 
Erklärung für die Zeit-Achse:     0s — 1s   Eindringen 
                            1s — 7s   Abrollen um 30° 

7s —8s    Verlassen 
 
Gruppe 2: d1, R1, Fw2, EN GJS-800-8 

 
Abb.44 

Gruppe 3: d1, R1, Fw3, EN GJS-800-8 
 

 
Abb.45 

 
 
 
Gruppe 4: d1, R1, Fw4, EN GJS-800-8 
 

 
Abb.46 

In Tabelle 7 sind die Eigenspannungen für die verschiedenen Beansprungsfälle 
aufgeführt. 

Tabelle 7 
  SX SY SZ 

d1,R1,Fw1 -70.15 -362.25 -477.13 
d1,R1,Fw2 -197.47 -821.50 -717.21 
d1,R1,Fw3 -256.77 -689.37 -326.58 
d1,R1,Fw4 -220.04 -558.77 -314.79 
    
d1,R2,Fw1 -50.54 -249.75 -403.26 
d1,R2,Fw2 -39.83 -280.40 -438.51 
d1,R2,Fw3 -44.31 -305.82 -278.23 
d1,R2,Fw4 -75.62 -534.16 -256.45 
    
d2,R1,Fw1 -146.06 -458.20 -612.94 
d2,R1,Fw2 -236.03 -592.44 -739.40 
d2,R1,Fw3 -333.80 -712.73 -723.75 
d2,R1,Fw4 -222.19 -530.20 -590.10 
    
d2,R2,Fw1 -88.57 -402.16 -540.35 
d2,R2,Fw2 -101.12 -375.61 -580.06 
d2,R2,Fw3 -181.01 -462.32 -677.04 

 
 
 
 
 
 
 
 
 
EN 
GJS
-800
-8 

d2,R2,Fw4 -180.50 -586.61 -554.91 
     

d1,R1,Fw1 -85.29 -419.92 -533.37 
d1,R1,Fw2 70.17 -382.28 -492.73 
d1,R1,Fw3 -75.85 -662.48 -575.14 
d1,R1,Fw4 -166.73 -821.33 -539.00 
    
d1,R2,Fw1 -71.58 -307.13 -452.99 
d1,R2,Fw2 -82.20 -233.70 -541.11 
d1,R2,Fw3 -42.31 -260.82 -520.20 
d1,R2,Fw4 -53.12 -523.87 -587.41 
    
d2,R1,Fw1 -128.53 -575.37 -653.44 
d2,R1,Fw2 -326.00 -890.67 -1092.73 
d2,R1,Fw3 -202.07 -637.12 -889.43 
d2,R1,Fw4 50.75 -462.13 -609.02 
    
d2,R2,Fw1 -64.23 -453.88 -546.31 
d2,R2,Fw2 -96.14 -495.69 -669.88 
d2,R2,Fw3 -241.07 -706.51 -951.92 

 
 
 
 
 
 
 
 
EN 
GJS
-100
0-5 

d2,R2,Fw4 -245.87 -697.50 -957.42 

8.2 Aufgabe 2  

Tiefenprofil für Beanspruchungsfälle an einer Probenstelle bei 20°. 

 Abb.47: Tiefenprofil 
 
Nach einer feineren Verneztung der Probe können mehre Knoten für das Erstellen 
eines Tiefenprofils der entstandenen Druckeigenspannungen verwendet werden. Die 
Daten wurden in Excel verarbeitet. 
 
Mit den Ergebnisen der acht Beanspruchungfälle können die Effekte des 
Festwalzprozesses bei verschiedenen Parametern miteinander verglichen werden. 
 
Fall 1: d1, R1, Fw2, EN GJS-800-8 

 Abb.48 

Fall 2: d1, R2, Fw2, EN GJS-800-8 

 Abb.49 

 

Fall 3: d2, R1, Fw2, EN GJS-800-8 

 Abb.50 

 
Fall 4: d2, R2, Fw2, EN GJS-800-8 

 Abb.51 

Fall 5: d1, R2, Fw1, EN GJS-800-8 

 Abb.52 

 
Fall 6: d1, R2, Fw3, EN GJS-800-8 

 Abb.53 

 
Fall 7: d1, R2, Fw4, EN GJS-800-8 

 Abb.54 

Fall 8: d1, R2, Fw2, EN GJS-1000-5 

 Abb.55 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

    



 V 

ⅡⅡⅡⅡ Das ANSYS - Modell 
 
!------------------------------------------------------------------------------------------------------------------------- 

FINI 

/CLE 

/FILNAME,Festwalz 

/TITLE,FEM-Simulation des Festwalzprozesses 

/PREP7    ! Pre-Processing 

!=======PARAMETER==================================================== 

PI=ACOS(-1) 

 

!Konstruktive Parameter 

d=25          !Probendurchmesser: d1=25mm, d2=30mm ,d3=35mm 

R=1.5         !Radius: R1=1,1mm R2=1,5mm 

 

!Festwalzparameter 

DR=15         !Rollendurchmesser 

rR=1.45      !Profilradius der Walzrolle:  für R1, rR1=1,1mm 

              !                             für R2, rR2=1,45mm 

t=0.195       ! Eindringtiefe 

outres=2 

 

!------ Material ---------------------------------- --------------------------------------------------------------------- 

/INPUT,ADI_800,inp ! ADI_800 und ADI_1000  

MP,MU,1,0.15           

MP,EX,2,2.1E05    ! Young's modulus  

MP,NUXY,2,0.3   ! Poison's Constant  

MP,EX,3,2.1E05    ! E-Modul für Material Nr. 3 (Kontaktelemente) 

MP,NUXY,3,0.3     ! nu 

MP,MU,3,0           ! Reibwert 

 

ET,1,45 

 

! ======= Geometriemodellierung ============================================= 

!Probe 

 

b=sqrt(R*R-(R-0.4)*(R-0.4))  !hilfswert 

 

x1=0.1*d 

y1=d/2-x1 

 

K,1,2.5*d,d/2 

K,2,3.5*d-R-b,d/2 



 VI 

K,3,3.5*d-R,d/2-0.4 

K,4,3.5*d,d/2-0.4+R 

K,5,3.5*d,1.4*d/2 

K,6,4.5*d,1.4*d/2 

K,7,4.5*d,0 

K,8,2.5*d,0 

K,9,3.5*d-R-b,0 

K,10,3.5*d+x1,0 

K,11,3.5*d+x1,d/2-0.4+R 

K,12,3.5*d-R-b,y1 

K,13,4.5*d,d/2-0.4+R 

K,14,3.5*d+x1,1.4*d/2 

K,15,2.5*d,y1 

K,20,3.5*d-R,d/2-0.4+R 

 

LARC,2,3,20,R 

LARC,3,4,20,R 

LCOMB,1,2 

 

A,2,12,11,4 

A,12,11,10,9 

A,4,5,14,11 

A,13,6,14,11 

A,11,13,7,10 

A,1,2,12,15 

A,8,9,12,15 

ALLSEL 

 

!Walzrolle 

 

LOCAL,11,0,3.5*d-R,d/2-0.4+DR/2,0,-90,0 

CSYS,11 

K,30,0,0 

K,31,0,rR 

K,32,DR/2-rR,rR 

K,33,DR/2,0 

K,34,DR/2-rR,-rR 

K,35,0,-rR 

K,36,DR/2-rR,0 

K,37,DR/2-2*rR,0 

LARC,32,33,36,rR 

LARC,33,34,36,rR 

LCOMB,21,22 

A,31,32,37,30 



 VII  

A,35,34,37,30 

A,32,37,34 

 

!-----------Netzung(Probe)------------------------------------------------------------------------------------------ 

EZ1=6 

EZ2=4 

EZ3=2 

EZ4=6 

EZ5=4 

EZ6=4 

EZ7=2 

 

LESIZE,1,,,EZ1 

LESIZE,3,,,EZ1,3 

LESIZE,6,,,EZ1,1/3 

 

LESIZE,2,,,EZ2 

LESIZE,4,,,EZ2 

LESIZE,9,,,EZ2 

LESIZE,18,,,EZ2 

 

LESIZE,8,,,EZ3,2 

LESIZE,10,,,EZ3,1/2 

LESIZE,11,,,EZ3,2 

 

LESIZE,5,,,EZ4,1.5 

LESIZE,7,,,EZ4,1/1.5 

LESIZE,14,,,EZ4,1.5 

LESIZE,20,,,EZ4,1.5 

 

LESIZE,12,,,EZ5,1/3 

LESIZE,13,,,EZ5,3 

LESIZE,15,,,EZ5,1/3 

 

LESIZE,16,,,EZ6,1/3 

LESIZE,17,,,EZ6,3 

LESIZE,19,,,EZ6,1/3 

 

VROTAT,1,2,3,,,,10,7,30 

VROTAT,4,5,6,7,,,10,7,30 

 

VROTAT,15,18,22,,,,10,7,60 

VROTAT,27,30,35,38,,,10,7,60 

VGLUE,all 



 VIII  

LESIZE,99,,,EZ5,3 

LESIZE,101,,,EZ5,3 

LESIZE,103,,,EZ5,3 

LESIZE,105,,,EZ5,3 

 

LESIZE,100,,,EZ6,3 

LESIZE,102,,,EZ6,3 

LESIZE,104,,,EZ6,3 

LESIZE,106,,,EZ6,3 

 

MAT,1 

type,1 

MSHAPE,0 

MSHKEY,1 

VMESH,all 

ALLSEL 

 

!------------Netzung(Walzrolle)------------------------------------------------------------------------------------ 

REZ1=6 

REZ2=6 

 

LESIZE,21,,,REZ1 

LESIZE,24,,,REZ2 

VROTAT,10,9,8,,,,31,35,360 

 

MAT,2 

type,1 

MSHAPE,0 

MSHKEY,1 

VSEL,all 

VSEL,s,,,4,7,1 

VSEL,a,,,11,14,1 

VSEL,a,,,23,26,1 

VMESH,all 

!------------------------------------------------------------------------------------------------------------------------- 

WSORT   ! Optimierung der Element- und Knotennummern 

!------------------------------------------------------------------------------------------------------------------------- 

! KONTACT - DEFINITION 

MAT,3   

R,3,,,0.5,0.2,0.001,,,,(SIGMAF*0.577),0.0,1.0,,    

REAL,3    

ET,2,170   ! Target - Elements target170 

ET,3,173   ! Contact 

 



 IX  

KEYOPT,3,1,  ! 1 - only structural contact 

KEYOPT,3,2,0  ! 2 - Algoritmus: 0 - Penalty function + Lagrange multiplier  

KEYOPT,3,3,0  ! 3 - Use with h-elements (no superelements) 

KEYOPT,3,4,0   ! 4 - Location of contact detection point: 0 - Gauss-Point  

KEYOPT,3,5,  ! 5 - CNOF Automated adjustment 

KEYOPT,3,6,  ! 6 - Symmetrical stiffness matrix 

KEYOPT,3,7,0  ! 7 - Element level time inc. control.; 0 - No control 

KEYOPT,3,8,0   ! 8 - Spurious Contact Prevention 0 - no prevention 

KEYOPT,3,9,0  ! 9 - include both init. - geom. penet. or gap and offset 

KEYOPT,3,10,0  !10 - Contact stiffness update: 0 - no update 

KEYOPT,3,11,0  !11 - Beam/Shell Thickness Effect: 0 - None  

KEYOPT,3,12,0  !12 - Behavior of contact surface: 0 - Standart        

 

!----------------Contact Fläche-------------------------------------------------------------------------------------- 

ALLSEL         

ASEL,S,,,26    ! Contact Fläche 

NSLA,S,1  

ESLN,R,0 

TYPE,3 

REAL,3 

ESURF,ALL  

ALLSEL 

   

!---------------Target Fläche---------------------------------------------------------------------------------------- 

ALLSEL         

ASEL,S,,,14     ! Target Fläche 

NSLA,S,1  

ESLN,R,0 

TYPE,2 

REAL,3 

ESURF,ALL  

ALLSEL 

 

!-------------Randbedingunen-------------------------------------------------------------------------------------- 

csys,0 

ASEL,S,,,1,7,1 

NSLA,S,1  

D,ALL,UZ,0 

NSEL,S,LOC,X,4.5*d    ! Einspannung der Probe 

D,ALL,ux,0 

NSEL,S,LOC,Y,0 

D,ALL,UY,0 

allsel 

 



 X 

!------------------------------------------------------------------------------------------------------------------------- 

LOCAL,12,1,3.5*d-R,0,0,0,0,-90 

csys,12 

NSEL,s,,,2101 

NSEL,a,,,2106 

NSEL,a,,,2711 

NSEL,a,,,2111 

NSEL,a,,,3245 

NROTAT,all 

allsel 

 

finish 

!====================================================================== 

/SOLU 

ANTYPE,STATIC 

 

!---------Eindringen------------------------------- ------------------------------------------------------------------ 

TIME,1 

rr=DISTND(2101,2106) 

R1=d/2-0.4 

R2=DR/2 

R4=R1+R2-t 

R3=R1+R2 

Psi=(atan(rr/R4)-atan(rr/R3))*180/pi 

 

D,2101,ux,-t 

D,2101,uy,0 

D,2101,uz,0 

D,2106,ux,-t 

D,2106,uy,0 

D,2106,uz,0 

D,2711,ux,-t 

D,2711,uy,0 

D,2711,uz,0 

D,3245,ux,-(sqrt(rr*rr+R3*R3)-sqrt(rr*rr+R4*R4)*cos (Psi*pi/180)) 

D,3245,uy,sqrt(rr*rr+R4*R4)*sin(Psi*pi/180) 

D,3245,uz,0 

D,2111,ux,-(sqrt(rr*rr+R3*R3)-sqrt(rr*rr+R4*R4)*cos (Psi*pi/180)) 

D,2111,uy,-sqrt(rr*rr+R4*R4)*sin(Psi*pi/180) 

D,2111,uz,0 

allsel 

NLGEOM,ON   ! nonlinear geometry 

AUTOTS,ON 

NSUBST,500,500000,50 



 XI 

NROPT,FULL,,ON 

KBC,0       ! Randbedingung (Eindringen) erfolgt über eine 

Rampe 

OUTRES,ALL,outres ! Ergebnisse von jedem 10-ten step 

Solve 

 

!---------Abrollen--------------------------------- -------------------------------------------------------------------- 

Phi_Ink=5 

Schritte=30/Phi_Ink 

 

*DO,k,1,Schritte 

 

    Time,k+1 

     

    !Parameter fur Mittelpunkt 

    Phi0=Phi_Ink*k 

    Ur0=R4 

     

    !Parameter fur punkt1 

    Alpha=(R4-R2)*Phi0/R2 

    Theta1=Alpha-Phi0 

    Ur1=sqrt(rr*rr+Ur0*Ur0-2*rr*Ur0*cos(Theta1*pi/1 80)) 

    Beta1=acos((Ur1*Ur1+Ur0*Ur0-rr*rr)/(2*Ur1*Ur0)) *180/pi 

    Phi1=Phi0-Beta1 

     

    !Parameter fur punkt2 

    Theta2=180-Theta1 

    Ur2=sqrt(rr*rr+Ur0*Ur0-2*rr*Ur0*cos(Theta2*pi/1 80)) 

    Beta2=acos((Ur2*Ur2+Ur0*Ur0-rr*rr)/(2*Ur2*Ur0)) *180/pi 

    Phi2=Phi0+Beta2 

     

    !Parameter fur punkt3 

    Theta3=90+Theta1 

    Ur3=sqrt(rr*rr+Ur0*Ur0-2*rr*Ur0*cos(Theta3*pi/1 80)) 

    Beta3=acos((Ur3*Ur3+Ur0*Ur0-rr*rr)/(2*Ur3*Ur0)) *180/pi 

    Phi3=Phi0-Beta3+atan(rr/R3)*180/pi 

     

    !Parameter fur punkt4 

    Theta4=90-Theta1 

    Ur4=sqrt(rr*rr+Ur0*Ur0-2*rr*Ur0*cos(Theta4*pi/1 80)) 

    Beta4=acos((Ur4*Ur4+Ur0*Ur0-rr*rr)/(2*Ur4*Ur0)) *180/pi 

    Phi4=Phi0+Beta4-atan(rr/R3)*180/pi 

    !Mittelpunkt 

    D,2101,ux,-(R3-Ur0*cos(Phi0*pi/180)) 

    D,2101,uy,-Ur0*sin(Phi0*pi/180) 

     

    !Punkt1 

    D,2106,ux,-((R3-rr)-Ur1*cos(Phi1*pi/180)) 

    D,2106,uy,-Ur1*sin(Phi1*pi/180) 

 

    !Punkt2 

    D,2711,ux,-((R3+rr)-Ur2*cos(Phi2*pi/180)) 

    D,2711,uy,-Ur2*sin(Phi2*pi/180) 

 

    !Punk3 

    D,3245,ux,-(sqrt(rr*rr+R3*R3)-Ur3*cos(Phi3*pi/1 80)) 

    D,3245,uy,-Ur3*sin(Phi3*pi/180) 

 

    !Punkt4 

    D,2111,ux,-(sqrt(rr*rr+R3*R3)-Ur4*cos(Phi4*pi/1 80)) 

    D,2111,uy,-Ur4*sin(Phi4*pi/180) 

 

    allsel 

    NLGEOM,ON  

    AUTOTS,ON 

    NSUBST,500,500000,50 

    NROPT,FULL,,ON 

    KBC,0  

    OUTRES,ALL,outres  

    solve 

     

*ENDDO 

 

!---------Verlassen------------------------------------------------------- 

Time,k+2 

 

!Parameter fur Mittelpunkt 

 

Phi0=Phi_Ink*k 

Ur0=R3 

 

!Parameter fur punkt1 

    Alpha=(R4-R2)*Phi0/R2 

    Theta1=Alpha-Phi0 

    Ur1=sqrt(rr*rr+Ur0*Ur0-2*rr*Ur0*cos(Theta1*pi/1 80)) 

    Beta1=acos((Ur1*Ur1+Ur0*Ur0-rr*rr)/(2*Ur1*Ur0)) *180/pi 

    Phi1=Phi0-Beta1 

    !Parameter fur punkt2 

    Theta2=180-Theta1 

    Ur2=sqrt(rr*rr+Ur0*Ur0-2*rr*Ur0*cos(Theta2*pi/1 80)) 

    Beta2=acos((Ur2*Ur2+Ur0*Ur0-rr*rr)/(2*Ur2*Ur0)) *180/pi 

    Phi2=Phi0+Beta2 

     

    !Parameter fur punkt3 

    Theta3=90+Theta1 

    Ur3=sqrt(rr*rr+Ur0*Ur0-2*rr*Ur0*cos(Theta3*pi/1 80)) 

    Beta3=acos((Ur3*Ur3+Ur0*Ur0-rr*rr)/(2*Ur3*Ur0)) *180/pi 

    Phi3=Phi0-Beta3+atan(rr/R3)*180/pi 

     

    !Parameter fur punkt4 

    Theta4=90-Theta1 

    Ur4=sqrt(rr*rr+Ur0*Ur0-2*rr*Ur0*cos(Theta4*pi/1 80)) 

    Beta4=acos((Ur4*Ur4+Ur0*Ur0-rr*rr)/(2*Ur4*Ur0)) *180/pi 

    Phi4=Phi0+Beta4-atan(rr/R3)*180/pi 

 

    !Mittelpunkt 

    D,2101,ux,-(R3-Ur0*cos(Phi0*pi/180)) 

    D,2101,uy,-Ur0*sin(Phi0*pi/180) 

     

    !Punkt1 

    D,2106,ux,-((R3-rr)-Ur1*cos(Phi1*pi/180)) 

    D,2106,uy,-Ur1*sin(Phi1*pi/180) 

 

    !Punkt2 

    D,2711,ux,-((R3+rr)-Ur2*cos(Phi2*pi/180)) 

    D,2711,uy,-Ur2*sin(Phi2*pi/180) 

 

    !Punk3 

    D,3245,ux,-(sqrt(rr*rr+R3*R3)-Ur3*cos(Phi3*pi/1 80)) 

    D,3245,uy,-Ur3*sin(Phi3*pi/180) 

 

    !Punkt4 

    D,2111,ux,-(sqrt(rr*rr+R3*R3)-Ur4*cos(Phi4*pi/1 80)) 

    D,2111,uy,-Ur4*sin(Phi4*pi/180) 

     

    allsel 

    NLGEOM,ON 

    AUTOTS,ON 

    NSUBST,500,500000,50 

    NROPT,FULL,,ON 

    KBC,0  

    OUTRES,ALL,outres 

    solve 

 

FINI 

 

!====================================================================== 

/POST1 

csys,12 

rsys,12 

 

!------------------------------------------------------------------------------------------------------------------------- 

! Aufgabe 1 

!------------------------------------------------------------------------------------------------------------------------- 

*GET,ns,ACTIVE,0,SOLU,NCMSS ! Anzahl der Zeitschritte ermitteln --> ns 

 

!------ Definition von Ausgabevektoren ------------------------------------------------------------------------- 

*DIM,C1,ARRAY,(8*ns/outres) ! Zaehler 

*DIM,NSX,ARRAY,(8*ns/outres) ! radiale Spannung 

*DIM,NSY,ARRAY,(8*ns/outres) ! tangentiale Spannung 

*DIM,NSZ,ARRAY,(8*ns/outres) ! axiale Spannung 

 

*DO,i,1,(8*ns/outres) 

 C1(i)=i*(outres/ns)               ! Anfangswerte setzten 

*ENDDO 

!------------------------------------------------------------------------------------------------------------------------- 

ALLSEL 

*DO,j,1,8    

*DO,i,1,(ns/outres)  ! loop for every 10th substep 

 

 SET,j,outres*i 

 zl=i+ns/outres*(j-1) 

 GN=NODE(d/2-0.4,75,0) 

 *GET,NSX_zl,NODE,GN,S,X   ! 

 *GET,NSY_zl,NODE,GN,S,Y   ! 

 *GET,NSZ_zl,NODE,GN,S,Z   ! 

   

 NSX(zl)=NSX_zl 

 NSY(zl)=NSY_zl 

 NSZ(zl)=NSZ_zl 

          

*ENDDO 

*ENDDO 

ALLSEL 

!------------------------------------------------------------------------------------------------------------------------- 

*CREATE,WriteRes.mac 

/OUT,Ergebnis,dat 

/COM, 

/COM, 

/COM,------------------------------------------------------------------------------------------------------------------ 

/COM,|        c1        |        SX        |        SY        |        SZ        |   

/COM,------------------------------------------------------------------------------------------------------------------ 

*VWRITE,C1(1),NSX(1),NSY(1),NSZ(1) 

(' |',F15.5,'   |',F15.5,'   |',F15.5,'   |',F15.5,'   |') 

/COM,------------------------------------------------------------------------------------------------------------------ 

/OUT 

*END 

/INPUT,WriteRes.mac  

*LIST,Ergebnis,dat 

 

!------------------------------------------------------------------------------------------------------------------------- 

! Aufgabe 2 : Tiefenprofil 

!------------------------------------------------------------------------------------------------------------------------- 

SET,8,LAST      ! letzter Schritt (t = 7s) 

csys,12 

rsys,12 

 

*DIM,NSX,ARRAY,(8) ! Druckeigenspannungen 

*DIM,Tiefe,ARRAY,(8) ! Abtragstiefen 

ALLSEL 

 

*GET,NSX(1),NODE,162,S,X   ! 

*GET,NSX(2),NODE,295,S,X   ! 

*GET,NSX(3),NODE,300,S,X   ! 

*GET,NSX(4),NODE,305,S,X   ! 

*GET,NSX(5),NODE,310,S,X   ! 

*GET,NSX(6),NODE,315,S,X   ! 

*GET,NSX(7),NODE,83,S,X   ! 

*GET,NSX(8),NODE,537,S,X   ! 

 

Tiefe(1)=0 

Tiefe(2)=(NX(162)+UX(162))-(NX(295)+UX(295)) 

Tiefe(3)=(NX(162)+UX(162))-(NX(300)+UX(300)) 

Tiefe(4)=(NX(162)+UX(162))-(NX(305)+UX(305)) 

Tiefe(5)=(NX(162)+UX(162))-(NX(310)+UX(310)) 

Tiefe(6)=(NX(162)+UX(162))-(NX(315)+UX(315)) 

Tiefe(7)=(NX(162)+UX(162))-(NX(83)+UX(83)) 

Tiefe(8)=(NX(162)+UX(162))-(NX(537)+UX(537)) 

ALLSEL 

!------------------------------------------------------------------------------------------------------------------------- 

*CREATE,WriteRes.mac 

/OUT,Ergebnis,dat 

/COM, 

/COM, 

/COM,------------------------------------------------------------------------------------------------------------------ 

/COM,|       Tiefe      |        SX        | 

/COM,------------------------------------------------------------------------------------------------------------------ 

*VWRITE,Tiefe(1),NSX(1) 

(' |',F15.5,'   |',F15.5,'   |') 

/COM,------------------------------------------------------------------------------------------------------------------ 

/OUT 

*END 

/INPUT,WriteRes.mac  

*LIST,Ergebnis,dat 

 
!------------------------------------------------------------------------------------------------------------------------- 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



 XVII  

ⅢⅢⅢⅢ Literaturverzeichnis 
 
[1] www.ecoroll.de, 09.09.2009, Startseite > Verfahren > Vorteile:  „Grundlagen 
des Festwalzens“ 
 
[2] www. baublies.de, 09.09.2009, Startseite > Werkzeuge > 
Einrollenwerkzeuge,variabel > Innenkonturen 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 



 XVIII  

ⅣⅣⅣⅣ Eidesstattliche Erklärung 

 

Ich erkläre, daß ich die vorliegende Arbeit selbständig, ohne fremde 

Hilfe angefertigt und nur die in den beigefügten Verzeichnissen 

angegebenen Hilfsmittel verwendet habe. 

 

[Ort], den [Datum]       [Unterschrift] 

 


